

Appendix 1. The example of NONMEM control file of semi-compartmental model.

```

$PROBLEM PD with Ke0 fit
$DATA
$INPUT
$PRED
  E0 = THETA(1);Baseline
  EMAX = THETA(2);Max Effect
  CE50 = THETA(3)*EXP(ETA(1));C50, only parameter with
          ;interindividual variability
  GAM = THETA(4); Gamma - don't add an ETA to this - very hard to fit
  KE0 = THETA(5)*EXP(ETA(2)); Ke0

  IF (TIME.EQ.0) THEN
    CE = 0
    PTIME = 0
    PCP = 0
    PCE = 0
  ENDIF

  DT = TIME-PTIME;

  IF (DT.EQ.0) THEN; Fighting with problem of nested if statements
    DT1 = 1
  ELSE
    DT1 = DT
  ENDIF

  IF (CP.GE.PCP) THEN
    SLOPE = (CP-PCP)/DT1
    DELTA = DT1*SLOPE+(KE0*PCP-SLOPE)*(1-EXP(-KE0*DT1))/KE0
  ELSE
    SLOPE = (LOG(CP)-LOG(PCP))/DT1
    DELTA = PCP*KE0/(KE0+SLOPE)*(EXP(DT1*SLOPE)-EXP(-KE0*DT1))
  ENDIF

  IF (DT.GT.0) THEN
    CE = PCE*EXP(-KE0*DT)+DELTA
  ELSE
    CE = PCE
  ENDIF

; Full sigmoidal Emax PD model with simple additive error
Y = E0+(EMAX-E0)*CE**GAM/(CE**GAM+C50**GAM) + EPS(1)

; Variables for NONMEM to remember at the next iteration
PTIME = TIME; Prior Time
PCP = CP; Prior plasma concentration
PCE = CE; Prior effect site concentration

$ESTIMATION SIG=3 MAX=2000 PRINT=1 METHOD = 1; FOCE method

```